

Contents: Basic course G3 and Upgrade course TNC 410

No.	Title	PGM-No.
	Cartesian contour movements	
1	Holes	151
2	Square	152
3	Rounding /chamfering corners	153
4	Rounding corners	154
5	Linear movements	250
6	Circular movements	251
7	Circular arc with CC, C	206
8	Tangential contour connection	207
9	Circular arcs	208
10	Circular arc with CR	209
	Polar contour movements	
11	Hexagon (polar)	213
12	Circle (polar) CP	211
13	Circular path with tang. connection (polar) CTP	212
14	Polar coordinates (general)	252
	Canned cycles	
15	Drilling cycle	201
16	Drilling with 200-series cycles	260
17	Slot plate	210
18	Bushing plate	262
19	Die I	265
20	Linear hole pattern	220
21	Hole pattern	221
22	Guide plate	261
23	Die II	266
	Program section repeats	
24	Hexagon	268
25	Drilled plate - slanted rows	270





Contents: Basic course G3 and Upgrade course TNC 410

No.	Title	PGM-No.
	Free Contour Programming	
26	FK Cam	288
27	Maltese cross	275
28	FK Hook - type spanner wrench	295
	Subprogramming Technique	
29	Subprogram (Groups of holes)	215
30	Bolt hole circle segments with several tools	280
31	Milling with several settings	223
	Contour Cycles	
32	Contour cycles SL 1, Roughing out kidney	174
	Coordinate Transformations	
33	Datum shift and mirror images	229
34	Coordinate transformations (combined)	232
35	Scaling factor - hammer	284
36	Scaling factor	234



Task: **Holes**

Program(s): _____



Program layout: **Holes**

Begin program

Define workpiece blank

Define tool

Call tool data

Move to clearance height

Move to starting point

Move to set up clearance

Drill

Retract drill

Move to next position

Drill















Retract drill

Next hole


Drill

Return to clearance height

End of program

	<i>BEGINN PGM ... MM</i>
	<i>BLK-FORM 0.1 ... X... Y... Z...</i> <i>BLK-FORM 0.2 X... Y... Z...</i>
	<i>TOOL DEF ... L ... R ...</i>
	<i>TOOL CALL ... S ...</i>
	<i>L Z+100 R0 FMAX</i>
	<i>L X... Y... R0 FMAX M3</i>
	<i>L Z+... R FMAX M8</i>
	<i>L Z-... R F100 M</i>
	<i>L Z+2 R0 FMAX</i>
	<i>L X... Y... FMAX</i>
	<i>L Z-...</i>
	<i>L Z+... FMAX</i>
	<i>L X... Y... R0 FMAX</i>
	<i>L Z-22 R0</i>
	<i>L Z+100 R0 FMAX M2 (M30)</i>





Complete program


```

0 BEGIN PGM 151 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0..... WORKPIECE BLANK DEFINITION
3 TOOL DEF 1 L+0 R+4 ..... DEFINE TOOL
4 TOOL CALL 1 Z S4000 ..... CALL TOOL DATA
5 L Z+100 R0 F MAX ..... CLEARANCE HEIGHT
6 L X+20 Y+30 R0 F MAX M3 ..... FIRST HOLE
7 L Z+2 R0 F MAX M8
8 L Z-22 R0 F400 ..... DRILL
9 L Z+2 R0 F MAX
10 L X+50 Y+70 R0 F MAX ..... SECOND HOLE
11 L Z-22 R0 F400
12 L Z+2 R0 F MAX
13 L X+75 Y+30 R0 F MAX ..... THIRD HOLE
14 L Z-22 R0 F400
15 L Z+100 R0 F MAX M2
16 END PGM 151 MM

```

Task: **Square**

Program(s): _____



Begin program
Define workpiece blank

Define tool
Call tool data
Move to clearance height
Move to auxiliary point R0
Plunging depth
Contour starting point RL/RR
Contour coordinates RL/RR
⋮
Last contour point RL/RR
Move to auxiliary point R0
Retract tool, PGM end

```
BEGIN PGM... MM  
BLK-FORM 0.1... X... Y... Z...  
BLK-FORM 0.2 X... Y... Z...  
TOOL DEF... L... R...  
TOOL CALL... S...  
L Z+... R0 F...  
  
L X... Y... R... F...  
  
L Z... R... F... M...  
  
L X... Y... RR/RL  
  
L X... Y... RR/RL  
⋮  
L X... Y... RR/RL  
  
L X... Y... R...  
  
L Z... R... F... M...
```




HEIDENHAIN

Basic course G3/Upgrade course D02

Solution:

Square



Complete program

```
0 BEGIN PGM 152 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0 ..... WORKPIECE BLANK DEFINITION
3 TOOL DEF 1 L+0 R+8 ..... DEFINE TOOL
4 TOOL CALL 1 Z S4000 ..... CALL TOOL DATA
5 L Z+100 R0 F MAX ..... CLEARANCE HEIGHT
6 L X-30 Y+50 R0 F MAX ..... AUXILIARY POINT (R0)
7 L Z-5 R0 F MAX M3 ..... PLUNGING DEPTH
8 L X+0 Y+50 RL F400 ..... CONTOUR START. POINT (RL/RR)
9 L X+50 Y+100
10 L X+100 Y+50
11 L X+50 Y+0
12 L X+0 Y+50 RL ..... LAST CONTOUR POINT
13 L X-30 R0 F MAX M5 ..... AUXILIARY POINT
14 L Z+100 R0 F MAX M2 ..... RETRACT TOOL/PGM END
15 END PGM 152 MM
```




HEIDENHAIN

Basic course G3/Upgrade course C01


Task: **Rounding / chamfering corners**

Program(s): _____



Solution:


Rounding / chamfering corners



Complete program

```
0 BEGIN PGM 153 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0 ..... WORKPIECE BLANK DEFINITION
3 TOOL DEF 1 L+0 R+8 ..... DEFINE TOOL
4 TOOL CALL 1 Z S4000 ..... CALL TOOL DATA
5 L Z+100 R0 F MAX ..... CLEARANCE HEIGHT
6 L X-30 Y+50 R0 F MAX ..... AUXILIARY POINT (R0)
7 L Z-5 R0 F MAX M3
8 L X+0 Y+50 RL F200 ..... CONTOUR STARTING POINT RL
9 L X+50 Y+100
10 RND R10 ..... ROUND CORNERS
11 L X+100 Y+50
12 L X+50 Y+0
13 CHF 5 ..... CHAMFER
14 L X+0 Y+50 RL
15 L X-30 R0 M5 ..... AUXILIARY POINT (R0)
16 L Z+100 R0 F MAX M2 ..... PGM END
17 END PGM 153 MM
```






- Begin program
- Define workpiece blank
- Define tool
- Call tool data
- Move to clearance height
- Auxiliary point anfahren
- Plunging depth
- Approach tangentially
- Contour
- Depart tangentially
- Retract tool, PGM end

```

BEGIN PGM... MM
BLK-FORM 0.1... X... Y... Z...
BLK-FORM 0.2 X... Y... Z...
TOOL DEF... L... R...
TOOL CALL..... S...
L Z...
L X... Y...
L Z...
APPR...
L...
DEP...
L Z...
    
```

Solution:

Rounding corners



Complete program

```
0 BEGIN PGM 154 MM
1 BLK FORM 0.1 Z X-20 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+8
4 TOOL CALL 1 Z S4000
5 L Z+100 R0 F MAX
6 L X-30 Y+70 R0 F MAX ..... AUXILIARY POINT (R0)
7 L Z-5 R0 F MAX M3
8 APPR LCT X+10 Y+70 R5 RL F400 ..... APPROACH STARTING POINT
TANGENTIALLY



9 L X+10 Y+90
10 RND R10
11 L X+50 Y+90
12 L Y+50 X+90
13 L X+90 Y+10
14 RND R10
15 L X+50 Y+10
16 L X+10 Y+50
17 L Y+70 ..... LAST CONTOUR POINT RL
18 DEP LCT X-30 Y+70 R5 R0 ..... DEPART TANGENTIALLY TO
AUXILIARY POINT


19 L Z+100 R0 F MAX M2
20 END PGM 154 MM
```



Task: **Linear movements**

Program(s): _____





Complete program



```


0 BEGIN PGM 250 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0 ..... WORKPIECE BLANK DEFINITION
3 TOOL CALL 7 Z S2500 ..... CALL TOOL DATA; R4
4 L Z+100 R0 F9999 ..... CLEARANCE HEIGHT
5 L X-20 Y+40 ..... AUXILIARY POINT (R0)
6 L Z+2 M3
7 L Z-5 F1000
8 APPR LCT X+5 Y+40 R3 RL F300 ..... CONTOUR STARTING POINT
9 L Y+90 (APPROACH TANGENTIALLY)
10 CHF 20
11 L X+60
12 L X+90 Y+80
13 L Y+40
14 CHF 10
15 L X+60 Y+10
16 L X+5
17 RND R10
18 L Y+40 ..... LAST CONTOUR POINT
19 DEP LCT X-20 Y+40 R3 R0 ..... AUXILIARY POINT (R0)
20 L Z+100 R0 F MAX M2 ..... RETRACT TOOL
21 END PGM 250 MM
    
```



Task: **Circular movements**

Program(s): _____






Complete program

```

0 BEGIN PGM 251 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 7 Z S2500 ..... R4
4 L Z+100 R0 F9999
5 L X+20 Y-20 ..... AUXILIARY POINT (R0)
6 L Z+2 M3
7 L Z-5 F500
8 APPR LCT X+20 Y+30 R3 RL F300 ..... CONTOUR STARTING POINT
9 L X+0 (APPROACH TANGENTIALLY)
10 RND R4
11 L X+15 Y+45
12 CR X+15 Y+60 R+20 DR+
13 L X+0 Y+75
14 CR X+20 Y+95 R+20 DR-
15 L X+40
16 CT X+65 Y+80
17 CC X+75 Y+80
18 C X+85 Y+80 DR+
19 L X+95
20 RND R5
21 L Y+50
22 L X+75 Y+30
23 RND R8
24 L Y+20
25 CC X+60 Y+20
26 C X+45 Y+20 DR-
27 L Y+30
28 RND R9
29 L X+20 ..... LAST CONTOUR POINT
30 DEP LCT X+20 Y-20 R3 R0 F500 ..... AUXILIARY POINT (R0)
31 L Z+100 R0 F MAX M2
32 END PGM 251 MM
    
```




- Begin program
- Define workpiece blank
- Call tool data
- Move to clearance height
- Move to auxiliary point R0
- Plunging depth
- Approach contour tangentially
- Define circle center
- Circular movement
- Depart tangentially (aux.pt.)
- Retract tool, PGM end

```

BEGIN PGM... MM
BLK-FORM 0.1... X... Y... Z...
BLK-FORM 0.2 X... Y... Z...
TOOL CALL... S...
L...
L... R0
L...
APPR... RL/RR
CC...
C...
DEP...
L...
    
```


Solution:

Circular arc with CC, C





Complete program

```
0 BEGIN PGM 206 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 L X-30 Y+50 R0 F MAX ..... AUXILIARY POINT
6 L Z-5 R0 F MAX M3
7 APPR LT X+0 Y+20 LEN10 RL F250 M8
8 L X+23,542 RL
9 CC X+50 Y+50 ..... CIRCLE CENTER
10 C Y+80 X+23,542 DR+ ..... CIRCULAR MOVEMENT
11 L X+0 RL
12 DEP LT LEN10 R0 ..... DEPART TANGENTIALLY (AUX.PT.)
13 L Z+100 R0 F MAX M2
14 END PGM 206 MM
```




Task: **Tangential contour connection
(cartesian)**

Program(s): _____



Solution:

Tangential contour connection (cartesian)




Complete program

```
0 BEGIN PGM 207 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 11 Z S2500 ..... R10
4 L Z+100 R0 F MAX ..... CLEARANCE HEIGHT
5 L X+45 Y-25 R0 F MAX ..... AUXILIARY POINT
6 L Z-5 F MAX M13
7 APPR LT X+20 Y+0 LEN5 RR F250 ..... APPROACH STARTING POINT OF
8 L Y+30 ..... CONTOUR TANGENTIALLY
9 CT X+70 Y+60 ..... TANGENTIAL CIRCULAR PATH
10 L X+100
11 DEP LT LEN5 R0
12 L Z+100 R0 F MAX M2
13 END PGM 207 MM
```



Task: **Circular arcs**

Program(s): _____




Point	X	Y
P ₁	6,645	35,495
P ₂	55,505	69,488

Point	X	Y
P ₃	58,995	30,025
P ₄	19,732	21,191

Solution:

Circular arcs



Complete program

```


0 BEGIN PGM 208 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX ..... CLEARANCE HEIGHT
5 L X-30 Y+30 R0 F MAX M3 ..... AUXILIARY POINT
6 L Z-5 F MAX ..... DEPTH
7 APPR LCT X+5 Y+30 R5 RL F250 M8 ..... APPROACH TANGENTIALLY
8 CC X+15 Y+30 ..... CIRCLE CENTER
9 C X+6,645 Y+35,495 DR- ..... CIRCLE
10 CT X+55,505 Y+69,488 ..... TANGENTIAL CIRCULAR PATH
11 CC X+60 Y+50
12 C X+58,995 Y+30,025 DR-
13 CT X+19,732 Y+21,191
14 CC X+15 Y+30
15 C X+5 Y+30 DR-
16 DEP LCT X-30 Y+30 R5 R0 ..... DEPART TANGENTIALLY
17 L Z+100 R0 F MAX M2
18 END PGM 208 MM

```



HEIDENHAIN

Basic course G3/Upgrade course D02




- Begin program
- Define workpiece blank
- Call tool data
- Move to clearance height
- Move to auxiliary point
- Plunging depth
- Approach tangentially
- Contour
- Depart tangentially
- Retract tool, PGM end

```

BEGIN PGM... MM
BLK-FORM 0.1... X... Y... Z...
BLK-FORM 0.2 X... Y... Z...
TOOL CALL... S...
L...
...
...
...
...
...
    
```

Solution:

Circular arc with CR




Complete program

```
0 BEGIN PGM 209 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 L X-30 Y-30 R0 F MAX M13 ..... AUXILIARY POINT
6 L Z-5 F MAX
7 APPR LT X+10 Y+0 LEN10 RL F250 ..... APPROACH STARTING POINT
TANGENTIALLY
8 L Y+50
9 CR X+90 Y+50 R+80 DR- ..... CIRCULAR ARC
10 L Y+0
11 DEP LT LEN10 R0 ..... DEPART TANGENTIALLY (AUX.PT.)
12 L Z+100 R0 F MAX M2
13 END PGM 209 MM
```



Task: **Hexagon (polar)**

Program(s): _____



- Begin program
- Define workpiece blank


- Call tool data
- Move to clearance height

- Define pole
- Auxiliary point in polar coord.
- Plunging depth
- Approach tangentially
- Contour
 - ⋮
- Depart tangentially
- Retract tool, PGM end

```
BEGIN PGM... MM  
BLK-FORM 0.1... X... Y... Z...  
BLK-FORM 0.2 X... Y... Z...  
TOOL CALL... S...  
L Z...  
  
CC X... Y...  
LP PR... PA...  
  
L Z...  
  
APPR LCT X... Y... R...  
  
LP PR... PA...  
⋮  
DEP LCT X... Y... R...  
L Z...
```


Solution:

Hexagon (polar)




Complete program

```
0 BEGIN PGM 213 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-20
2 BLK FORM 0.2 X+50 Y+50 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 CC X+0 Y+0 ..... POLE
6 LP PR+80 PA+0 R0 F MAX ..... AUXILIARY POINT (POLAR)
7 L Z-5 F MAX M3
8 APPR LCT X+40 Y+0 R5 RR F250 M8 ..... APPROACH STARTING POINT OF
9 LP PA+60 ..... CONTOUR TANGENTIALLY
10 LP PA+120
11 LP PA+180
12 LP PA+240
13 LP PA+300
14 LP PA+360
15 DEP LCT X+80 Y+0 R5 ..... DEPART TANGENTIALLY
16 L Z+100 R0 F MAX M2
17 END PGM 213 MM
```



Task: **Circle (polar) CP**

Program(s): _____




- Begin program
- Define workpiece blank
- Call tool data
- Move to clearance height
- Define pole
- Auxiliary point in polar coord.
- Contour
- Auxiliary point
- Retract tool, PGM end

```
BEGIN PGM ... MM  
BLK-FORM 0.1 ... X... Y... Z...  
BLK-FORM 0.2 X... Y... Z...  
TOOL CALL ... S...  
L...  
...  
...  
...  
...  
...
```

Solution:

Circle (polar) CP




Complete program

```
0 BEGIN PGM 211 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 CC X+50 Y+50 ..... POLE
6 LP PR+80 PA+180 R0 F MAX M3 ..... AUXILIARY POINT
7 L Z-5 F MAX
8 APPR LCT X+10 Y+50 R5 RL F250 M8 ..... APPROACH STARTING POINT OF
CONTOUR TANGENTIALLY
9 CP IPA+360 DR- ..... CIRCLE PATH (POLAR)
10 DEP LCT X-30 Y+50 R5 R0 ..... DEPART TANGENTIALLY
11 L Z+100 F MAX M2
12 END PGM 211 MM
```



Task: **Circular path with tangential connection (polar) CTP**

Program(s): _____




- Begin program
- Define workpiece blank
- Call tool data
- Move to clearance height
- Define pole
- Auxiliary point in polar coord.
- Contour
- Auxiliary point
- Retract tool, PGM end

```

BEGIN PGM ... MM
BLK-FORM 0.1 ... X... Y... Z...
BLK-FORM 0.2 X... Y... Z...
TOOL CALL ... S...
L ...
...
...
...
...
...
    
```

Solution:

Circular path with tangential connection (polar) CTP



Complete program

```


0 BEGIN PGM 212 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+70 Z+0
3 TOOL CALL 13 Z S4000 ..... R20
4 L Z+100 R0 F MAX
5 L X-30 Y-30 R0 F MAX M3 ..... AUXILIARY POINT
6 L Z-5 F MAX
7 APPR LT X+0 Y+0 LEN5 RL F250 M8 ..... APPROACH STARTING POINT OF
8 L X+0 Y+30 ..... CONTOUR TANGENTIALLY
9 L X+40 Y+60
10 CC Y+20 X+20 ..... POLE
11 CTP PR+76,158 PA+23,199 ..... TANGENTIAL CIRCULAR PATH
(POLAR)
12 L Y+0
13 DEPT LT LEN5 R0 ..... DEPART TANGENTIALLY
14 L Z+100 R0 F MAX M2
15 END PGM 212 MM

```




HEIDENHAIN

Basic course G3/Upgrade course D02



Solution:

Polar coordinates (general)



Complete program


```

0 BEGIN PGM 252 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 L X+130 Y+50 F MAX M3
6 L Z-5 F MAX
7 APPR LCT X+92 Y+50 R5 RL F250 M8 ..... APPROACH STARTING POINT OF
8 L Y+40 ..... CONTOUR TANGENTIALLY
9 CC X+92 Y+40 ..... POLE
10 LP PR+45 PA-135 ..... STRAIGHT LINE (POLAR)
11 CC X+92 Y+50
12 LP PR+90 PA-150
13 CP PA+148 DR- ..... CIRCULAR PATH (POLAR)
14 L IX+30 IY+0
15 CC ..... DEFINE AS POLE POSITION
16 LP PR+21 PA-60
17 CC X+92 Y+88
18 CTP PR+29 PA-90 ..... TANGENTIAL PATH (POLAR)
19 L Y+50
20 DEP LCT X+130 Y+50 R5 R0 ..... DEPART TANGENTIALLY
21 L Z+100 R0 F MAX M2
22 END PGM 252 MM
  
```



HEIDENHAIN

Basic course G3/Upgrade course D02



- Begin program
- Define workpiece blank
- Call tool data
- Define cycle
- Move to clearance height
- Starting point first hole / call cycle
- Second hole / call cycle
- Third hole / call cycle
- Retract tool, PGM end

```


BEGIN PGM ... MM
BLK-FORM 0.1 ... X... Y... Z...
BLK-FORM 0.2 X... Y... Z...
TOOL CALL ... S...
CYCL DEF ...
L Z+ ...
L X... Y... M99

L X... Y... M99

L X... Y... M99
...
    
```


Solution:



Drilling cycle



Complete program

```
0 BEGIN PGM 201 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 7 Z S1000 ..... R4
4 CYCL DEF 200 DRILLING
  Q200=2 ..... SET UP CLEARANCE
  Q201=-23 ..... DEPTH
  Q206=150 ..... FEED RATE FOR PLUNGING
  Q202=5 ..... PLUNGING DEPTH
  Q210=0 ..... DWELL TIME AT TOP
  Q203=+0 ..... SURFACE COORDINATE
  Q204=2 ..... 2ND SET UP CLEARANCE
5 L Z+100 R0 F9999 M3
6 L X+20 Y+30 M99 ..... STARTING POSITION
7 L X+50 Y+70 M99 ..... 2ND HOLE
8 L X+75 Y+30 M99 ..... 3RD HOLE
9 L Z+100 R0 F MAX M2
10 END PGM 201 MM
```





Procedure:

- Centering
- Drilling diameter 28 mm
- Boring
- Reaming
- Tapping

Cycle 1

Cycle 203


Cycle 202

Cycle 201

Cycle 2

Solution:

Drilling with 200-series cycles



Complete program

Centering

```
0 BEGIN PGM 260 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-70
2 BLK FORM 0.2 X+50 Y+50 Z+0
3 TOOL CALL 9 Z S1000 ..... R6
4 L Z+100 R0 F MAX
5 CYCL DEF 1.0 PECKING
6 CYCL DEF 1.1 SET UP 2
7 CYCL DEF 1.2 DEPTH -5
8 CYCL DEF 1.3 PECKG 5
9 CYCL DEF 1.4 DWELL 0
10 CYCL DEF 1.5 F200
11 L X+0 Y+0 R0 F9999 M3
12 L Z+2 M99
13 L Z+100 M6
```

Drilling

```
14 TOOL CALL 11 Z S350 ..... R10
15 CYCL DEF 203 UNIVERSAL DRILLING
    Q200=5 ..... SET UP CLEARANCE
    Q201=-80 ..... DEPTH
    Q206=100 ..... FEED RATE FOR PLUNGING
    Q202=15 ..... PLUNGING DEPTH
    Q210=0 ..... DWELL TIME AT TOP
    Q203=+0 ..... SURFACE COORDINATE
    Q204=20 ..... 2ND SET UP CLEARANCE
    Q212=2 ..... DECREMENT
    Q213=2 ..... NR OF BREAKS
    Q205=5 ..... MIN. PLUNGING DEPTH
    Q211=0 ..... DWELL TIME AT DEPTH
    Q208=500 ..... RETRACTION FEED RATE
16 CYCL CALL M3
17 L Z+100 M6
```



HEIDENHAIN

Basic course G3/Upgrade course D02



Solution: Drilling with 200-series cycles


Boring	18	TOOL CALL 13 Z S500	R20	
	19	CYCL DEF 202 BORING		
		Q200=2	SET UP CLEARANCE	
		Q201=-35	DEPTH	
		Q206=250	FEED RATE FOR PLUNGING	
		Q211=0	DWELL TIME AT DEPTH	
		Q208=500	RETRACTION FEED RATE	
		Q203=+0	SURFACE COORDINATE	
		Q204=20	2ND SET UP CLEARANCE	
		Q214=0	DISENGAGING DIRECTION	
		20	CYCL CALL M3	
	21	L Z+100 M6		
Reaming	22	TOOL CALL 13 Z S100	R20	
	23	CYCL DEF 201 REAMING		
		Q200=5	SET UP CLEARANCE	
		Q201=-25	DEPTH	
		Q206=100	FEED RATE FOR PLUNGING	
		Q211=0	DWELL TIME AT DEPTH	
		Q208=300	RETRACTION FEED RATE	
		Q203=+0	SURFACE COORDINATE	
		Q204=20	2ND SET UP CLEARANCE	
		24	CYCL CALL M3	
	25	L Z+100 M6		
Tapping	26	TOOL CALL 12 Z S50	R15	
	27	CYCL DEF 2.0 TAPPING		
	28	CYCL DEF 2.1 SET UP 5		
	29	CYCL DEF 2.2 DEPTH -45		
	30	CYCL DEF 2.3 DWELL 0		
	31	CYCL DEF 2.4 F100		
	32	L Z-30 M3		
	33	CYCL CALL		
		34	L Z+100 M2	
		35	END PGM 260 MM	



Task: **Slot plate**

Program(s): _____





Complete program

```


0 BEGIN PGM 210 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 8 Z S1000 ..... R5
4 L Z+100 R0 F9999 M3
5 CYCL DEF 210 SLOT RECIP. PLNG
  Q200=2 ..... SET UP CLEARANCE
  Q201=-10 ..... DEPTH
  Q207=100 ..... FEED RATE FOR MILLING
  Q202=5 ..... PLUNGING DEPTH
  Q215=0 ..... MACHINING OPERATION
  Q203=+0 ..... SURFACE COORDINATE
  Q204=2 ..... 2ND SET UP CLEARANCE
  Q216=+50 ..... CENTER IN 1ST AXIS
  Q217=+25 ..... CENTER IN 2ND AXIS
  Q218=75 ..... FIRST SIDE LENGTH
  Q219=15 ..... SECOND SIDE LENGTH
  Q224=+20 ..... ANGLE OF ROTATION
6 CYCL CALL

7 CYCL DEF 211 CIRCULAR SLOT
  Q200=2 ..... SET UP CLEARANCE
  Q201=-10 ..... DEPTH
  Q207=250 ..... FEED RATE FOR MILLING
  Q202=5 ..... PLUNGING DEPTH
  Q215=0 ..... MACHINING OPERATION
  Q203=+0 ..... SURFACE COORDINATE
  Q204=2 ..... 2ND SET UP CLEARANCE
  Q216=+40 ..... CENTER IN 1ST AXIS
  Q217=+60 ..... CENTER IN 2ND AXIS
  Q244=50 ..... PITCH CIRCLE DIAMETER
  Q219=12 ..... SECOND SIDE LENGTH
  Q245=+35 ..... STARTING ANGLE
  Q248=180 ..... ANGULAR LENGTH
8 CYCL CALL

9 L Z+100 M2
10 END PGM 210 MM
    
```

Task: **Bushing plate**

Program(s): _____



Program layout: **Bushing plate**

Conventional preparation: *BLK- FORM*

First tool Rough out circular pocket *TOOL CALL ...
CYCL DEF ...
... M99* (Oversizing DR!)

Rough out first long slot *CYCL DEF ...
CYCL CALL*

Rough out second long slot *FN 0: Q ...
CYCL CALL
... M6* Tool change

Second tool Finish first short slot *TOOL CALL ...
CYCL DEF ...
CYCL CALL*

Finish second short slot *FN 0: Q ...
CYCL CALL*

Finish first long slot *CYCL DEF ...
CYCL CALL*

Finish second long slot *FN 0: Q ...
CYCL CALL*

First rounded slot *CYCL DEF ...
CYCL CALL*

Second rounded slot *FN 0: Q ...
CYCL CALL*


Finish circular pocket *CYCL DEF ...
CYCL CALL*

Retract tool, end *L Z100 M2*



Solution:

Bushing plate



Complete program

```
0 BEGIN PGM 262 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-15
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 6 Z S1500 DR+0,5 ..... R3
4 L Z+100 R0 F9999
5 L X+50 Y+50 M3
6 CYCL DEF 5.0 CIRCULAR POCKET
7 CYCL DEF 5.1 SET UP2
8 CYCL DEF 5.2 DEPTH -15
9 CYCL DEF 5.3 PECKG 5 F100
10 CYCL DEF 5.4 RADIUS 10
11 CYCL DEF 5.5 F200 DR+
12 L Z+2 M99
13 CYCL DEF 210 SLOT RECIP. PLNG
    Q200=2 ..... SET UP CLEARANCE
    Q201=-6 ..... DEPTH
    Q207=200 ..... FEEDRATE FOR MILNG
    Q202=6 ..... PLUNGING DEPTH
    Q215=0 ..... MACHINING OPERATION
    Q203=+0 ..... SURFACE COORDINATE
    Q204=20 ..... 2ND SET UP CLEARANCE
    Q216=+50 ..... CENTER IN 1ST AXIS
    Q217=+15 ..... CENTER IN 2ND AXIS
    Q218=81 ..... FIRST SIDE LENGTH
    Q219=11 ..... SECOND SIDE LENGTH
    Q224=+0 ..... ANGLE OF ROTATION
14 CYCL CALL
15 FN 0: Q217 = +85
16 CYCL CALL
17 L Z+100 M6
```



HEIDENHAIN


Basic course G3/Upgrade course D02


18	TOOL CALL 5 Z S2000	R2,5
19	CYCL DEF 210 SLOT RECIP. PLNG	
	Q200=2	SET UP CLEARANCE.
	Q201=-9	DEPTH
	Q207=200	FEEDRATE FOR MILNG
	Q202=5	PLUNGING DEPTH
	Q215=0	MACHINING OPERATION
	Q203=-6	SURFACE COORDINATE
	Q204=20	2ND SET UP CLEARANCE
	Q216=+50	CENTER IN 1ST AXIS
	Q217=+15	CENTER IN 2ND AXIS
	Q218=76,6	FIRST SIDE LENGTH
	Q219=6,6	SECOND SIDE LENGTH
	Q224=+0	ANGLE OF ROTATION
20	CYCL CALL M3	
21	FN 0: Q217 = +85	
22	CYCL CALL	
23	CYCL DEF 210 SLOT RECIP. PLNG	
	Q200=2	SET UP CLEARANCE
	Q201=-6,8	DEPTH
	Q207=200	FEED RATE FOR MILLNG
	Q202=6,8	PLUNGING DEPTH
	Q215=0	MACHINING OPERATION
	Q203=+0	SURFACE COORDINATE
	Q204=20	2ND SET UP CLEARANCE
	Q216=+50	CENTER IN 1ST AXIS
	Q217=+15	CENTER IN 2ND AXIS
	Q218=81	FIRST SIDE LENGTH
	Q219=11	SECOND SIDE LENGTH
	Q224=+0	ANGLE OF ROTATION
24	CYCL CALL	
25	FN 0: Q217 = +85	
26	CYCL CALL	
27	CYCL DEF 211 CIRCULAR SLOT	
	Q200=2	SET UP CLEARANCE
	Q201=-15	DEPTH
	Q207=200	FEEDRATE FOR MILLING
	Q202=5	PLUNGING DEPTH
	Q215=0	MACHINING OPERATION
	Q203=+0	SURFACE COORDINATE
	Q204=20	2ND SET UP CLEARANCE
	Q216=+50	CENTER IN 1ST AXIS
	Q217=+50	CENTER IN 2ND AXIS
	Q244=60	PITCH CIRCLE DIAMETER
	Q219=6,6	SECOND SIDE LENGTH
	Q245=-30	STARTING ANGLE
	Q248=60	ANGULAR LENGTH
28	CYCL CALL	
29	FN 0: Q245 = +150	
30	CYCL CALL	
31	CYCL DEF 214 C. POCKT FINSHNG	
	Q200=2	SET UP CLEARANCE
	Q201=-15	DEPTH
	Q206=250	FEED RATE FOR PLUNGING
	Q202=8	PLUNGING DEPTH
	Q207=250	FEEDRATE FOR MILLING
	Q203=+0	SURFACE COORDINATE
	Q204=20	2ND SET UP CLEARANCE
	Q216=+50	CENTER IN 1ST AXIS
	Q217=+50	CENTER IN 2ND AXIS
	Q222=19	WORKPIECE BLANK DIAMETER
	Q223=20,02	FINISHED PART DIAMETER
32	CYCL CALL	
33	L Z+100 R0 F MAX M2	
34	END PGM 262 MM	



Task: **Die I**

Program(s): _____





Complete program

```

0 BEGIN PGM 265 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 9 Z S500 DR+1 ..... R6
4 L Z+100 R0 F9999
5 CC X+50 Y+50
6 LP PR+70 PA+0
7 L Z+2 M3
8 L Z-10 F200
9 CP PA+360 DR-
10 LP PR+60
11 CP PA+360 DR-
12 LP PR+45 RL
13 CP PA+360 DR-
14 LP PR+65 PA+0 R0
15 L Z+2
16 CYCL DEF 4.0 POCKET MILLING
17 CYCL DEF 4.1 SET UP 2
18 CYCL DEF 4.2 DEPTH -5
19 CYCL DEF 4.3 PECKG 5 F100
20 CYCL DEF 4.4 X60
21 CYCL DEF 4.5 Y30
22 CYCL DEF 4.6 F250 DR- RADIUS 7
23 L X+50 Y+50 R0 F MAX M99
24 L Z+100 F MAX M6

```





25	TOOL CALL 8 Z S1000	R5
26	CYCL DEF 212 POCKET FINISHING	
	Q200=2	SET UP CLEARANCE
	Q201=-5	DEPTH
	Q206=250	FEED RATE FOR PLUNGING
	Q202=5	PLUNGING DEPTH
	Q207=250	FEED RATE FOR MILLNG
	Q203=+0	SURFACE COORDINATE
	Q204=20	2ND SET UP CLEARANCE
	Q216=+50	CENTER IN 1ST AXIS
	Q217=+50	CENTER IN 2ND AXIS
	Q218=60	FIRST SIDE LENGTH
	Q219=30	SECOND SIDE LENGTH
	Q220=5,25	CORNER RADIUS
	Q221=0	ALLOWANCE IN 1ST AXS
27	CYCL CALL M3	
28	CYCL DEF 215 C. STUD FINISHNG	
	Q200=2	SET UP CLEARANCE
	Q201=-10	DEPTH
	Q206=250	FEED RATE FOR PLUNGING
	Q202=10	PLUNGING DEPTH
	Q207=250	FEED RATE FOR MILLNG
	Q203=+0	SURFACE COORDINATE
	Q204=20	2ND SET UP CLEARANCE
	Q216=+50	CENTER IN 1ST AXIS
	Q217=+50	CENTER IN 2ND AXIS
	Q222=92	WORKPIECE BLANK DIAMETER
	Q223=90	FINISHED PART DIAMETER
29	CYCL CALL	
30	L Z+100 R0 F MAX M2	
31	END PGM 265 MM	




Task: Linear hole pattern

Program(s): _____



Solution:

Linear hole pattern



Main program

```

0 BEGIN PGM 220 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 6 Z S1500 ..... R3
4 L Z+100 R0 F9999 M3
5 CYCL DEF 200 DRILLING
  Q200=2 ..... SET UP CLEARANCE
  Q201=-20 ..... DEPTH
  Q206=100 ..... FEED RATE FOR PLUNGING
  Q202=5 ..... PLUNGING DEPTH
  Q210=0 ..... DWELL TIME AT TOP
  Q203=+0 ..... SURFACE COORDINATE
  Q204=2 ..... 2ND SET UP CLEARANCE
6 CALL LBL 1

Retract tool, end      7 L Z+100 M2
  
```

SPGM

```

8 LBL 1
9 CYCL DEF 221 CARTESIAN PATTRN
  Q225=+15 ..... STARTNG PNT 1ST AXIS
  Q226=+20 ..... STARTNG PNT 2ST AXIS
  Q237=+10 ..... SPACING IN 1ST AXIS
  Q238=+8 ..... SPACING IN 2ND AXIS
  Q242=8 ..... NUMBER OF COLUMNS
  Q243=9 ..... NUMBER OF ROWS
  Q224=+0 ..... ANGLE OF ROTATION
  Q200=2 ..... SET UP CLEARANCE
  Q203=+0 ..... SURFACE COORDINATE
  Q204=2 ..... 2ND SET UP CLEARANCE
10 LBL 0
11 END PGM 220 MM
  
```





HEIDENHAIN

Basic course G3/Upgrade course D02


Task: **Hole pattern**

Program(s): _____



Solution:

Hole pattern



Main program

```

0 BEGIN PGM 221 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 6 Z S1500 ..... R3
4 L Z+100 R0 F9999 M3
5 CYCL DEF 200 DRILLING
  Q200=2 ..... SET UP CLEARANCE
  Q201=-17 ..... DEPTH
  Q206=150 ..... FEED RATE FOR PLUNGING
  Q202=6 ..... PLUNGING DEPTH
  Q210=0 ..... DWELL TIME AT TOP
  Q203=+0 ..... SURFACE COORDINATE
  Q204=2 ..... 2ND SET UP CLEARANCE
6 CALL LBL 1
7 CALL LBL 2

Retract tool, end      8 L Z+100 M2
  
```

SPGM

```

9 LBL 1
10 CYCL DEF 220 POLAR PATTERN
  Q216=+90 ..... CENTER IN 1ST AXIS
  Q217=+10 ..... CENTER IN 2ND AXIS
  Q244=100 ..... PITCH CIRCLE DIA.
  Q245=+90 ..... STARTING ANGLE
  Q246=+180 ..... STOPPING ANGLE
  Q247=15 ..... STEPPING ANGLE
  Q241=7 ..... NR OF REPETITIONS
  Q200=2 ..... SET UP CLEARANCE
  Q203=+0 ..... SURFACE COORDINATE
  Q204=2 ..... 2ND SET UP CLEARANCE
11 LBL 0
  
```



HEIDENHAIN

Basic course G3/Upgrade course D02

Solution:



Hole pattern

```
12 LBL 2
13 CYCL DEF 221 CARTESIAN PATTRN
    Q225=+30 ..... STARTNG PNT 1ST AXIS
    Q226=+30 ..... STARTNG PNT 2ND AXIS
    Q237=+15 ..... SPACING IN 1ST AXIS
    Q238=+20 ..... SPACING IN 2ND AXIS
    Q242=5 ..... NUMBER OF COLUMNS
    Q243=2 ..... NUMBER OF ROWS
    Q224=+80 ..... ANGLE OF ROTATION
    Q200=2 ..... SET UP CLEARANCE
    Q203=+0 ..... SURFACE COORDINATE
    Q204=2 ..... 2ND SET UP CLEARANCE
14 LBL 0
15 END PGM 221 MM
```



Task: Guide plate

Program(s): _____




- Procedure:**
- Shoulders
 - Centering
 - Drilling
 - Tapping

Cycle 200
Cycle 200
Cycle 2

Solution:

Guide plate



Main program

```
0 BEGIN PGM 261 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-30
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S500 ..... R20
4 L Z+100 R0 F9999
5 L X-50 Y-50
6 L Z+2 M3
7 L Z-10 F200
8 APPR LCT X+30 Y+0 R2 RL
9 L Y+100
10 DEP LCT X+0 Y+150 R2 R0
11 L X+100 F MAX
12 APPR LCT X+70 Y+100 R2 RL
13 L Y+0
14 DEP LCT X+100 Y-50 R2 R0
15 L Z+100 R0 F9999 M6

16 TOOL CALL 4 Z S2000 ..... R2,5
17 CYCL DEF 200 DRILLING
    Q200=2 ..... SET UP CLEARANCE
    Q201=-3,5 ..... DEPTH
    Q206=250 ..... FEED RATE FOR PLUNGING
    Q202=3,5 ..... PLUNGING DEPTH
    Q210=0 ..... DWELL TIME AT TOP
    Q203=-10 ..... SURFACE COORDINATE
    Q204=20 ..... 2ND SET UP CLEARANCE
18 L Z+5 M3
19 CALL LBL 1
20 L Z+100 M6
```



HEIDENHAIN

Basic course G3/Upgrade course D02

21	TOOL CALL 5 Z S2000	R2,5
22	CYCL DEF 200 DRILLING	
	Q200=2	SET UP CLEARANCE
	Q201=-15	DEPTH
	Q206=250	FEED RATE FOR PLUNGING
	Q202=5	PLUNGING DEPTH
	Q210=0	DWELL TIME AT TOP
	Q203=-10	SURFACE COORDINATE
	Q204=20	2ND SET UP CLEARANCE
23	L Z+5 M3	
24	CALL LBL 1	
25	L Z+100 M6	
26	TOOL CALL 6 Z S300	R3
27	CYCL DEF 2.0 TAPPING	
28	CYCL DEF 2.1 SET UP 2	
29	CYCL DEF 2.2 DEPTH -12	
30	CYCL DEF 2.3 DWELL 0	
31	CYCL DEF 2.4 F300	
32	L Z+5 M3	
33	CALL LBL 1	
<hr/>		
Retract tool, end	34 L Z+100 R0 F MAX M2	


SPGM

35	LBL 1	
36	CYCL DEF 221 CARTESIAN PATTRN	
	Q225=+15	STARTNG PNT 1ST AXIS
	Q226=+10	STARTNG PNT 2ND AXIS
	Q237=+70	SPACING IN 1ST AXIS
	Q238=+20	SPACING IN 2ND AXIS
	Q242=2	NUMBER OF COLUMNS
	Q243=5	NUMBER OF LINES
	Q224=+0	ANGLE OF ROTATION
	Q200=2	SET UP CLEARANCE
	Q203=-10	SURFACE COORDINATE
	Q204=20	2ND SET UP CLEARANCE
37	LBL 0	
38	END PGM 261 MM	

SPGM, end

Task: **Die II**

Program(s): _____




Program layout: **Die II**


Define workpc. blank	<i>BLK FORM</i>	
	<i>CALL PGM...</i>	PGM-Call
Drilling \varnothing 5 mm	<i>CYCL DEF...</i>	
	<i>CALL LBL...</i>	SPGM1-Call
	<i>L Z100... M6</i>	Tool change
Centering, pitch circle	<i>CYCL DEF...</i>	
	<i>CALL LBL...</i>	SPGM2-Call
	<i>L Z100... M6</i>	Tool change
Drilling, pitch circle	<i>CYCL DEF...</i>	
	<i>CALL LBL...</i>	SPGM2-Call
	<i>L Z100... M6</i>	Tool change
Tapping	<i>CYCL DEF...</i>	
	<i>CALL LBL...</i>	SPGM2-Call
Retract tool, end	<i>L Z... M2</i>	

SPGM1	<i>LBL 1</i>
Cartesian pattern	<i>CYCL DEF...</i>
	<i>LBL 0</i>
SPGM2	<i>LBL 2</i>
Polar pattern	<i>CYCL DEF...</i>
	<i>LBL 0</i>



Copy help program from PGM 265





Main program

```

0 BEGIN PGM 266 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 CALL PGM 26501 ..... HELP PROGRAM
4 TOOL CALL 5 Z S2000 ..... R2,5
5 L Z+100 M3
6 CYCL DEF 200 DRILLING
  Q200=2 ..... SET UP CLEARANCE
  Q201=-6 ..... DEPTH
  Q206=200 ..... FEED RATE FOR PLUNGING
  Q202=6 ..... PLUNGING DEPTH
  Q210=0 ..... DWELL TIME AT TOP
  Q203=-5 ..... SURFACE COORDINATE
  Q204=20 ..... 2ND SET UP CLEARANCE
7 CALL LBL 1
8 L Z+100 R0 F9999 M6

9 TOOL CALL 4 Z S2000 ..... R2
10 L Z+100 M3
11 CYCL DEF 1.0 PECKING
12 CYCL DEF 1.1 SET UP 2
13 CYCL DEF 1.2 DEPTH -2,8
14 CYCL DEF 1.3 PECKG 2,8
15 CYCL DEF 1.4 DWELL 0
16 CYCL DEF 1.5 F200
17 CALL LBL 2
18 L Z+100 M6

19 TOOL CALL 5 Z S1500 ..... R2,5
20 L Z+100 M3
21 CYCL DEF 200 DRILLING
  Q200=2 ..... SET UP CLEARANCE
  Q201=-15 ..... DEPTH
  Q206=200 ..... FEED RATE FOR PLUNGING
  Q202=8 ..... PLUNGING DEPTH
  Q210=0 ..... DWELL TIME AT TOP
  Q203=+0 ..... SURFACE COORDINATE
  Q204=20 ..... 2ND SET UP CLEARANCE
22 CALL LBL 2
23 L Z+100 M6
    
```




```

24 TOOL CALL 6 Z S300 ..... R3
25 L Z+100 M3
26 CYCL DEF 2.0 TAPPING
27 CYCL DEF 2.1 SET UP 2
28 CYCL DEF 2.2 DEPTH -12
29 CYCL DEF 2.3 DWELL 1
30 CYCL DEF 2.4 F300
31 CALL LBL 2

32 L Z+100 R0 F MAX M2

```

SPGM

```

33 LBL 1
34 CYCL DEF 221 CARTESIAN PATTRN
   Q225=+30 ..... STARTNG PNT 1ST AXIS
   Q226=+42 ..... STARTNG PNT 2ND AXIS
   Q237=+10 ..... SPACING IN 1ST AXIS
   Q238=+8 ..... SPACING IN 2ND AXIS
   Q242=5 ..... NUMBER OF COLUMNS
   Q243=3 ..... NUMBER OF LINES
   Q224=+0 ..... ANGLE OF ROTATION
   Q200=2 ..... SET UP CLEARANCE
   Q203=-6 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
35 LBL 0


36 LBL 2
37 CYCL DEF 220 POLAR PATTERN
   Q216=+50 ..... CENTER IN 1ST AXIS
   Q217=+50 ..... CENTER IN 2ND AXIS
   Q244=80 ..... PITCH CIRCLE DIA.
   Q245=+0 ..... STARTING ANGLE
   Q246=+360 ..... STOPPING ANGLE
   Q247=30 ..... STEPPING ANGLE
   Q241=12 ..... NR OF REPETITIONS
   Q200=2 ..... SET UP CLEARANCE
   Q203=+0 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
38 LBL 0
39 END PGM 266 MM

```




Task: **Hexagon**

Program(s): _____



Solution: **Hexagon**



Complete program

```

0 BEGIN PGM 268 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 12 Z S500 ..... R15
4 L Z+100 R0 F9999
5 CC X+50 Y+50 ..... POLE
6 LP PR+80 PA+0 R0 ..... AUXILIARY POINT
7 L Z-5 M3
8 APPR LCT X+90 Y+50 R2 RL F100 ..... STARTING POINT



9 LBL 1 ..... DEFINE LABEL
10 LP PR+40 IPA-60
11 CALL LBL 1 REP 5/5 ..... CALL LABEL WITH REPEATS

12 DEP LCT X+130 Y+50 R2 R0 F200 ..... AUXILIARY POINT
13 L Z+100 R0 F MAX M2
14 END PGM 268 MM


```

Task: Drilled plate – slanted columns

Program(s): _____



Move up and down columns (meandering)




Drilling pattern

<i>L X... Y...</i> <i>L Z...</i>	Absolute pre-position Set up clearance
<i>LBL 1</i>	Define label 1
<i>L IX... M99</i>	Cross over to first hole
<i>LBL 2</i>	Define label 2
<i>L IY... M99</i>	Move up the column
<i>CALL LBL 2 REP...</i>	
<i>L IX... M99</i>	Cross over two second column
<i>LBL 3</i>	Define label 2
<i>L IY... M99</i>	Move down the column
<i>CALL LBL 3 REP...</i> <i>CALL LBL 1 REP...</i>	Remaining groups

Solution:

Drilled plate – slanted columns



Main program

```
0 BEGIN PGM 270 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 L Z+100 R0 F MAX

4 TOOL CALL 5 Z S4000 ..... R2,5
5 CYCL DEF 200 DRILLING
  Q200=2 ..... SET UP CLEARANCE
  Q201=-15 ..... DEPTH
  Q206=250 ..... FEED RATE FOR PLUNGING
  Q202=8 ..... PLUNGING DEPTH
  Q210=0 ..... DWELL TIME AT TOP
  Q203=+0 ..... SURFACE COORDINATE
  Q204=20 ..... 2ND SET UP CLEARANCE

6 L X-10 Y+10 R0 F9999 M3
7 L Z+2

8 LBL 1
9 L IX+20 M99

10 LBL 2
11 L IX+3 IY+15 M99
12 CALL LBL 2 REP 4/4

13 L IX+10 M99

14 LBL 3
15 L IX-3 IY-15 M99
16 CALL LBL 3 REP 4/4



17 CALL LBL 1 REP 2/2

18 L Z+100 R0 F MAX M2
19 END PGM 270 MM
```




HEIDENHAIN

Basic course G3/Upgrade course D02



Solution:

FK Cam



Complete program

```
0 BEGIN PGM 288 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 10 Z S4000 ..... R7,5
4 L Z+100 R0 F MAX
5 L X-15 Y+35 R0 F MAX M3
6 L Z-5 R0 F MAX
7 APPR LCT X+5 Y+35 R5 RL F400
8 FC DR- R25 CCX+30 CCY+35
9 FLT
10 FCT DR- R15 CCX+50 CCY+75
11 FLT
12 FCT DR- R20 CCX+75 CCY+25
13 FLT
14 FCT X+5 Y+35 DR- R25 CCX+30 CCY+35
15 DEP LCT X-15 Y+35 R5
16 L Z+100 R0 F MAX M2
17 END PGM 288 MM
```





HEIDENHAIN

Basic course G3/Upgrade course D02


Task: **Maltese cross**

Program(s): _____



Solution:

Maltese cross



Main program

```
0 BEGIN PGM 275 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-20
2 BLK FORM 0.2 X+50 Y+50 Z+0
3 TOOL CALL 8 Z S250074 ..... R5
4 L Z+100 R0 F MAX
5 L Z+2 R0 F MAX M3
6 CALL LBL 10

7 LBL 5
8 CYCL DEF 10.0 ROTATION
9 CYCL DEF 10.1 IROT-90
10 CALL LBL 10
11 CALL LBL 5 REP 2/2

12 CYCL DEF 10.0 ROTATION
13 CYCL DEF 10.1 ROT+0

14 L Z+100 R0 F MAX M2
```


SPGM

```
15 LBL 10
16 CC X+0 Y+0
17 LP PR+70 PA+180 R0 F1000
18 L Z-10
19 APPR LCT X-50 Y+0 R2 RL F500
20 FPOL X+0 Y+0
21 FC DR- R50 CCX+0 CCY+0
22 FL AN+65 PDX+0 PDY+0 D48
23 FSELECT 4
24 FC DR+ R38 CCPR+75 CCPA+135
25 FSELECT 2
26 FL AN+25 PDX+0 PDY+0 D48
27 FSELECT 1
28 FC PR+50 PA+90 DR- R50 CCA+90
29 DEP LCT X+0 Y+90 R2 R0
30 LBL 0
31 END PGM 275 MM
```




HEIDENHAIN

Basic course G3/Upgrade course D02



Solution:

FK Hook-type spanner wrench



Complete program

```
0 BEGIN PGM 295 MM
1 BLK FORM 0.1 Z X+0 Y-35 Z-20
2 BLK FORM 0.2 X+160 Y+35 Z+0
3 TOOL CALL 8 Z S4000 ..... R5
4 L Z+100 R0 F MAX
5 L X-10 Y-10 R0 F MAX
6 L Z-5 R0 F MAX M3
7 APPR LCT X+0 Y+0 R2 RR F400
8 FL AN+0
9 FCT DR- R70
10 FCT DR+ R60 CCX+105 CCY+28
11 FSELECT 2
12 FCT DR+ R12
13 FCT DR+ R18 CCX+138 CCY-5
14 FLT LEN5 AN+170
15 FL X+134 AN-150
16 FSELECT 1
17 FL LEN4 IAN-90
18 FCT DR- R8
19 FCT DR- R60 CCX+105
20 FSELECT 1
21 FCT DR- R40 CCX+100
22 FCT Y+20 DR+ R5
23 FSELECT 2
24 FLT X+0 AN+180
25 FSELECT 1
26 L Y+0
27 DEP LCT X-10 Y-10 R2 R0
28 L Z+100 R0 FMAX M2
29 END PGM 295 MM
```




HEIDENHAIN

Basic course G3/Upgrade course D02

Task: **Subprogram
(Groups of holes)**

Program(s): _____



Begin program

<i>BEGIN PGM... MM</i>
<i>•</i>
<i>TOOL CALL...</i>
<i>CYCL DEF...</i>
<i>L Z+...</i>
<i>L X... Y...</i>
<i>L Z...</i>
<i>CALL LBL...</i>
<i>L X... Y...</i>
<i>CALL LBL...</i>
<i>L X... Y...</i>
<i>CALL LBL...</i>

- Call tool data
- Define cycle
- Move to clear. height
- Start. pos.-hole group ①
- Set up clearance
- Call SPGM
- Start. pos.-hole group ②
- Call SPGM
- Start. pos.-hole group ③
- Call SPGM

Retract tool, end

L Z +100 R0 F9999 M2

SPGM


<i>LBL...</i>

SPGM end

<i>LBL 0</i>

Solution:

Subprogram (Groups of holes)



Main program

```

0 BEGIN PGM 215 MM
1 BLK FORM 0.1 Z X+0 Y-70 Z-20
2 BLK FORM 0.2 X+100 Y+0 Z+0
3 TOOL CALL 4 Z S4000 ..... CALL TOOL DATA, R2
4 CYCL DEF 200 DRILLING
  Q200=2 ..... SET UP CLEARANCE
  Q201=-18 ..... DEPTH
  Q206=200 ..... FEED RATE FOR PLUNGING
  Q202=4 ..... PLUNGING DEPTH
  Q210=0 ..... DWELL TIME AT TOP
  Q203=+0 ..... SURFACE COORDINATE
  Q204=2 ..... 2ND SET UP CLEARANCE
5 L Z+100 R0 F MAX ..... CLEARANCE HEIGHT
6 L X+20 Y-10 R0 F MAX ..... STARTING POINT-HOLE GROUP ①
7 L Z+2 R0 F MAX M13
8 CALL LBL 1 ..... CALL SPGM

9 L X+63 Y-12 F MAX ..... STARTING POINT-HOLE GROUP ②
10 CALL LBL 1 ..... CALL SPGM

11 L X+42 Y-32 F MAX ..... STARTING POINT-HOLE GROUP ③
12 CALL LBL 1 ..... CALL SPGM

Retract tool, end
13 L Z+100 R0 F MAX M2

```

SPGM

```

14 LBL 1
15 CYCL CALL
16 L IX+15 F MAX M99
17 L IY-15 F MAX M99 ..... DRILLING PATTERN
18 L IX-15 F MAX M99
19 LBL 0
20 END PGM 215 MM

```

SPGM, end




HEIDENHAIN

Basic course G3/Upgrade course D02

Task: **Bolt hole circle segments with several tools**

Program(s): _____



Program layout:

Nesting for double bolt hole circles

Conventional Preparation:

BLK- FORM

Centering

*TOOL CALL ...
CYCL DEF / L Z+100*

CALL LBL 1

Drilling

*TOOL CALL ...
CYCL DEF ...*

CALL LBL 1

Tapping

*TOOL CALL ...
CYCL DEF ...*

CALL LBL 1

Retract tool, end

L Z100 M2

SPGM1

LBL 1

CC X... Y...

Circle center-left

CALL LBL 2

Call bolt hole circle seg.

CC X... Y...

Circle center-right

CALL LBL 2

Call bolt hole circle seg.

⋮

SPGM1 end

LBL 0

**SPGM2,
Bolt hole circle
segments**

LBL 2

*LP PR... PA... M3
L Z+2 M99*

Starting position

LBL 3

Remaining holes

⋮

CALL LBL 3 REP...

Partial repeat

LP PR... PA...

LBL 4

⋮

CALL LBL 4 REP...

SPGM2 end

LBL 0




HEIDENHAIN

Basic course G3/Upgrade course C03

Solution:

Bolt hole circle segments with several tools



Main program

```

0 BEGIN PGM 280 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+150 Y+60 Z+0
3 TOOL CALL 4 Z S2000 ..... R2
4 L Z+100 R0 F9999
5 CYCL DEF 1.0 PECKING
6 CYCL DEF 1.1 SET UP 2
7 CYCL DEF 1.2 DEPTH -3,5
8 CYCL DEF 1.3 PECKG 3,5
9 CYCL DEF 1.4 DWELL 0
10 CYCL DEF 1.5 F200
11 CALL LBL 1
12 L Z+100 M6

13 TOOL CALL 5 Z S1500 ..... R2,5
14 CYCL DEF 200 DRILLING
    Q200=2 ..... SET UP CLEARANCE
    Q201=-15 ..... DEPTH
    Q206=250 ..... FEED RATE FOR PLUNGING
    Q202=8 ..... PLUNGING DEPTH
    Q210=0 ..... DWELL TIME AT TOP
    Q203=+0 ..... SURFACE COORDINATE
    Q204=20 ..... 2ND SET UP CLEARANCE
15 CALL LBL 1
16 L Z+100 M6

17 TOOL CALL 6 Z S300 ..... R3
18 CYCL DEF 2.0 TAPPING
19 CYCL DEF 2.1 SET UP 2
20 CYCL DEF 2.2 DEPTH -12
21 CYCL DEF 2.3 DWELL 1
22 CYCL DEF 2.4 F300
23 CALL LBL 1

24 L Z+100 R0 M2
    
```



HEIDENHAIN

Basic course G3/Upgrade course D02

Solution:

Bolt hole circle segments with several tools

SPGM

```
25 LBL 1
26 CC X+40 Y+0
27 CALL LBL 2
28 CC X+110 Y+0
29 CALL LBL 2
30 LBL 0

31 LBL 2
32 LP PR+55 PA+120 R0 M3
33 L Z+2 M99

34 LBL 3
35 LP IPR-10 M99
36 CALL LBL 3 REP 3/3

37 LP PR+15 PA+60 R0 M99



38 LBL 4
39 LP IPR+10 M99
40 CALL LBL 4 REP 3/3

41 LBL 0
42 END PGM 280 MM
```




Task: **Milling with several settings**

Program(s): _____



Solution:

Milling with several settings



Main program

```
0 BEGIN PGM 223 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX M3
5 L X-30 Y+70 R0 F MAX ..... STARTING POSITION
6 L Z+0 F MAX

7 LBL 2
8 L IZ-5 R0 F MAX M3 ..... SETTING
9 CALL LBL 1 ..... CALL CONTOUR

10 CALL LBL 2 REP 5/5 ..... FURTHER CONTOUR SECTIONS

Retract tool, end 11 L Z+100 R0 F MAX M2
```

SPGM, Contour

```
12 LBL 1
13 APPR LCT X+10 Y+70 R5 RL F250 M3
14 L X+10 Y+90 RL
15 RND R10
16 L X+50 Y+90
17 RND R20
18 L X+90 Y+50
19 RND R20 ..... CONTOUR
20 L X+90 Y+10
21 RND R10
22 L X+50 Y+10
23 RND R20
24 L X+10 Y+50
25 RND R20
26 L X+10 Y+70
27 DEP LCT X-20 Y+70 R5 F500
28 LBL 0
29 END PGM 223 MM
```

SPGM end



HEIDENHAIN

Basic course G3/Upgrade course D02

Program layout: **SL-cycles I**

Conventional preparation

*BLK FORM
TOOL DEF
TOOL CALL 1 ...*

Workpiece blank
Tool list
First tool call

Process

List contour elements

*CYCL 14 CONTOUR GEOMETRY
CONTOUR LABEL 1/.../n*

Pilot drilling

*CYCL 15 PILOT DRILLING
L Z2 ... M3
CYCL CALL
L Z 100 ... M6*

Definition cycle 15
Clearance height
Call cycle
Tool change

Roughing-out

*TOOL CALL ...
CYCL 6 ROUGH-OUT
L Z2 ... M3
CYCL CALL
L Z100 ... M6*

Call tool data
Definition cycle 6
Clearance height
Call cycle
Tool change

Finishing

*TOOL CALL ...
CYCL 16 CONTOUR MILLING
L Z2 ... M3
CYCL CALL*

Process end

L Z100 ... M2

Contour elements, SPGM 1

*LBL 1
L X... Y... **RR**
L X... Y...*

No depth or feed rate values, no M-function, no approach and departure, with **radius correction**

LBL 0


**SPGM 2, SPGM 3, ...
SPGM n**

•
•
•
•
•
•



Solution:

Contour cycles SL 1 roughing out kidney



MAIN PROGRAM

```
0 BEGIN PGM 174 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+4
4 TOOL CALL 1 Z S4000
5 L Z+100 R0 F MAX M3
6 CYCL DEF 14.0 CONTOUR GEOMETRY
7 CYCL DEF 14.1 CONTOUR LABEL 1
8 CYCL DEF 6.0 ROUGH-OUT
9 CYCL DEF 6.1 SET-UP -2 DEPTH -18
10 CYCL DEF 6.2 PECKG -5 F500 ALLOW +0,5
11 CYCL DEF 6.3 ANGLE +10 F500
12 L X+30 Y+30 R0 F MAX M13 ..... PRE-POSITION
13 L Z+2 R0 F MAX M99 ..... CALL CYCLE
14 CYCL DEF 16.0 CONTOUR MILLING
15 CYCL DEF 16.1 SET-UP -2 DEPTH -18
16 CYCL DEF 16.2 PECKG -18 F200 DR+ F400
17 CYCL CALL

Retract tool, end      18 L Z+100 R0 F MAX M2
```



SPGM, Contour

```
19 LBL 1
20 L X+5 Y+30 RR
21 CC X+15 Y+30
22 C X+6,645 Y+35,495 DR-
23 CT X+55,505 Y+69,488
24 CT X+58,995 Y+30,025
25 CT X+19,732 Y+21,191
26 CC X+15 Y+30
27 C X+5 Y+30 DR-
28 LBL 0
29 END PGM 174 MM
```



Task: Datum shift and mirror images

Program(s): _____



BEGIN PGM ... MM

•
•
•
•
•
•
•
•
•
•
•
•
•
•
•
•
•
•
•
•
•
•

- Datum shift
- Part ① SPGM-Call
- Part ② Mirror images
SPGM-Call
- Part ③ Mirror images
SPGM-Call
- Part ④ Mirror images
SPGM-Call
- Reset cycle

Retract tool, end

L Z +100 R0 F9999 M2

SPGM

LBL ...

SPGM end

LBL 0




HEIDENHAIN

Basic course G3/Upgrade course C01

Solution:

Datum shift and mirror images



MAIN PROGRAM

```

0 BEGIN PGM 229 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 7 Z S4000 ..... R4
4 L Z+100 R0 F MAX
5 CYCL DEF 7.0 DATUM SHIFT
6 CYCL DEF 7.1 X+50
7 CYCL DEF 7.2 Y+50
8 CALL LBL 1

```

```

9 CYCL DEF 8.0 MIRROR IMAGE
10 CYCL DEF 8.1 X
11 CALL LBL 1

```

```

12 CYCL DEF 8.0 MIRROR IMAGE
13 CYCL DEF 8.1 Y
14 CALL LBL 1

```

```

15 CYCL DEF 8.0 MIRROR IMAGE
16 CYCL DEF 8.1 X Y
17 CALL LBL 1

```

Retract tool, end

```

18 L Z+100 R0 F MAX M2

```

SPGM, Contour

```

19 LBL 1
20 L X+0 Y+0 R0 F MAX M3 ..... AUXILIARY POINT R0
21 L Z+2 F MAX
22 L Z-15 R0 F100 ..... PLUNGING DEPTH
23 APPR LCT X+10 Y+10 R5 RL F200 ..... APPROACH STARTING POINT
                                         OF CONTOUR TANGENTIALLY
24 L Y+45
25 L X+30 Y+10
26 L X+10 Y+10
27 DEP LCT X+0 Y+0 R5 R0 ..... DEPART TANGENTIALLY
28 LBL 0
29 END PGM 229 MM

```






HEIDENHAIN

Basic course G3/Upgrade course D02

Task: **Coordinate transformations**

Program(s): _____





MAIN PROGRAM

```

0 BEGIN PGM 232 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 6 Z S4000 ..... R3
4 L Z+100 R0 F MAX M3
5 CYCL DEF 7.0 DATUM SHIFT
6 CYCL DEF 7.1 X+20
7 CYCL DEF 7.2 Y+10
8 CALL LBL 1

9 CYCL DEF 7.0 DATUM SHIFT
10 CYCL DEF 7.1 X+60
11 CYCL DEF 7.2 Y+10
12 CYCL DEF 11.0 SCALING
13 CYCL DEF 11.1 SCL1,2
14 CALL LBL 1

15 CYCL DEF 11.0 SCALING
16 CYCL DEF 11.1 SCL1
17 CYCL DEF 7.0 DATUM SHIFT
18 CYCL DEF 7.1 X+20
19 CYCL DEF 7.2 Y+60
20 CYCL DEF 10.0 ROTATION
21 CYCL DEF 10.1 ROT+20
22 CALL LBL 1

23 CYCL DEF 10.0 ROTATION
24 CYCL DEF 10.1 ROT+0
25 CYCL DEF 7.0 DATUM SHIFT
26 CYCL DEF 7.1 X+90
27 CYCL DEF 7.2 Y+60
28 CYCL DEF 8.0 MIRROR IMAGE
29 CYCL DEF 8.1 X
30 CALL LBL 1
    
```



Solution:

Coordinate transformations

```
31 CYCL DEF 8.0 ROTATION
32 CYCL DEF 8.1
33 CYCL DEF 7.0 DATUM SHIFT
34 CYCL DEF 7.1 X+0
35 CYCL DEF 7.2 Y+0
```

Retract tool, end

```
36 L Z+100 R0 F MAX M2
```


SPGM, Contour

```
37 LBL 1
38 L X+10 Y+10 R0 F MAX M3
39 L Z+2 F MAX
40 L Z-15 R0 F100
41 APPR LCT X+0 Y+10 R2 RR
42 L Y+20 X+0
43 RND R4
44 L X+20 Y+20
45 RND R4
46 L Y+0
47 CC X+15 Y+0
49 C X+10 Y+0 DR-
49 L X+0 Y+0
50 RND R4
51 L X+0 Y+10
52 DEP LCT X+10 Y+10 R2 R0
53 L Z+2 R0 F MAX
54 LBL 0
55 END PGM 232 MM
```




Task: **Scaling factor – hammer**

Program(s): _____



Solution:

Scaling factor – hammer



Main program

```
0 BEGIN PGM 284 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-10
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 6 Z S2000..... R3
4 CYCL DEF 7.0 DATUM SHIFT
5 CYCL DEF 7.1 X+5
6 CYCL DEF 7.2 Y+5
7 CALL LBL 1

8 CYCL DEF 7.0 DATUM SHIFT
9 CYCL DEF 7.1 X+95
10 CYCL DEF 7.2 Y+15
11 CYCL DEF 10.0 ROTATION
12 CYCL DEF 10.1 ROT+105
13 CALL LBL 1

14 CYCL DEF 10.0 ROTATION
15 CYCL DEF 10.1 ROT+0
16 CYCL DEF 7.0 DATUM SHIFT
17 CYCL DEF 7.1 X+20
18 CYCL DEF 7.2 Y+95
19 CYCL DEF 8.0 MIRROR IMAGE
20 CYCL DEF 8.1 Y
21 CYCL DEF 26.0 AXIS-SPEC. SCALING
22 CYCL DEF 26.1 X1,9 Y1 CCX+10 CCY+95
23 CALL LBL 1

24 CYCL DEF 26.0 AXIS-SPEC. SCALING
25 CYCL DEF 26.1 X1 Y1 CCX+0 CCY+0
26 CYCL DEF 8.0 MIRROR IMAGE
27 CYCL DEF 8.1
28 CYCL DEF 7.0 DATUM SHIFT
29 CYCL DEF 7.1 X+0
30 CYCL DEF 7.2 Y+0

31 L Z+100 R0 F MAX M2
```



HEIDENHAIN

Basic course G3/Upgrade course D02

Solution:

Scaling factor – hammer



SPGM

```
32 LBL 1
33 L X+7 Y+15 R0 F300
34 L Z-5
35 APPR LCT X+0 Y+15 R1 RR
36 L Y+25
37 L X+5 Y+40
38 L X+10 Y+25
39 L Y+20
40 L X+40
41 L Y+10
42 L X+10
43 L Y+0
44 L X+0
45 L Y+15
46 DEP LCT X+8 Y+15 R1 R0
47 L Z+2 R0
48 LBL 0
49 END PGM 284 MM
```




Task: **Scaling factor**

Program(s): _____



Solution:

Scaling factor



MAIN PROGRAM

```

0 BEGIN PGM 234 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 8 Z S4000 ..... R5
4 L Z+100 R0 F MAX M3
5 CALL LBL 1 ..... SPGM-CALL PART ①

6 CYCL DEF 7.0 DATUM SHIFT
7 CYCL DEF 7.1 X+50
8 CYCL DEF 7.2 Y+50
9 CYCL DEF 26.0 AXIS-SPEC. SCALING
10 CYCL DEF 26.1 X1,5 CCX+10
11 CALL LBL 1 ..... SPGM-CALL PART ②

12 CYCL DEF 7.0 DATUM SHIFT
13 CYCL DEF 7.1 X+0
14 CYCL DEF 7.2 Y+0
15 CYCL DEF 26.0 AXIS-SPEC. SCALING
16 CYCL DEF 26.1 X1 ..... CYCLE RESET

Retract tool, end      17 L Z+100 R0 F MAX M2
  
```

SPGM, Contour

```

18 LBL 1
19 L X+0 Y+0 R0 F MAX
20 L Z+2 F MAX
21 L Z-10 R0 F100 M8
22 APPR LT X+10 Y+10 LEN5 RL F200
23 L IY+20
24 L IX+21
25 L IX+4 IY-4
26 L Y+10
27 L X+10
28 DEP LT LEN5 R0
29 L X+0 Y+0 R0
30 L Z+2 R0 F MAX
31 LBL 0
32 END PGM 234 MM
  
```



HEIDENHAIN

Basic course G3/Upgrade course D02